Casual effect estimation for binary mediator M, Count/Poisson outcome Y,

with covariate C

MLR estimation

```
VARIABLE:
 NAMES = y x m c; ! c is a covariate, i.e., a control variable
 CATEGORICAL = m;
 COUNT = y; ! defaults to Poisson

MODEL:
        ! outcome regression:
 y ON m (beta1)
   x (beta2)
        c (beta3);
 [y] (beta0);
        ! mediator regression:
 m ON x (gamma1)
 c (gamma2);
 [m$1] (gamma0);   ! note: Mplus parameterizes the threshold, so the
           ! logit intercept is -gamma0 in the usual GLM sense

MODEL CONSTRAINT:
! astar1, astar0 are x variable values, here set to (2 0):
NEW(astar1*1 astar0*0
c_mean
  p11 p10 p01 p00
```

```
    rate1_1 rate0_1 rate1_0 rate0_0
           ey11 ey10 ey00
   tie de total);
astar1 = 2; ! 80th percentile in this example
astar0 = 0; ! 20th percentile in this example
c_mean = 66; ! mean of the covariate in this example

! Logit equation for mediator M:
! P(M=1|X=astar1), sign flip for threshold param:
p11 = 1/(1+exp(gamma0-gamma1*astar1+gamma2*c_mean));
! P(M=1|X=astar0):
p10 = 1/(1+exp(gamma0-gamma1*astar0+gamma2*c_mean));
p01 = 1 - p11;
p00 = 1 - p10;

! rates for Y at each M value and X level

! M=1, astar1:
rate1_1 = exp(beta0+beta1+beta2*astar1+beta3*c_mean);
! M=0, astar1:
rate0_1 = exp(beta0+beta2*astar1+beta3*c_mean);
! M=1, astar0:
rate1_0 = exp(beta0+beta1+beta2*astar0+beta3*c_mean);
! M=0, astar0:
rate0_0 = exp(beta0+beta2*astar0+beta3*c_mean);
```

```
! expectations for Y:

! Ey(astar1,astar1), Ey(astar1,astar0), Ey(astar0,astar0)

ey11 = rate1_1*p11 + rate0_1*p01;   ! Ey(astar1, astar1)

ey10 = rate1_1*p10 + rate0_1*p00;   ! Ey(astar1, astar0)

ey00 = rate1_0*p10 + rate0_0*p00;   ! Ey(astar0, astar0)


! causal effect formulas:

tie   = ey11 - ey10; ! also called tnie where n stands for natural

de    = ey10 - ey00; ! also called pnde

total = tie + de;
```